IDENTIFICATION

PRODUCT CODE: MAINDEC-15-D2AA-D(D)

PRODUCT NAME: PDP-15 ASR 33/35
TELETYPE TEST: PART1

DATE CREATED: DECEMBER 15, 1969

MAINTAINER: DIAGNOSTIC GROUP

AUTHOR: R. KOLLER

(9)

PDP 15
TTY

1. ABSTRACT

THE PDP-15, ASR33/35 TELETYPE TESTS PART 1, IS THE FIRST OF A 2 PART TEST PACKAGE USED TO TEST THE ASR33 OR ASR35 TELETYPE WHEN ATTACHED TO A PDP-15.

PART 1 CONTAINS SEVEN SELECTABLE PROGRAMS USED TO TEST THE INPUT AND OUTPUT LOGIC, AND THE TELETYPE READER. THE PROGRAMS ARE SELECTED AND CONTROLLED BY MEANS OF THE ACCUMULATOR SWITCHES (ACS).

THE PROGRAMS AVAILABLE ARE:

PRG0=BASIC INPUT LOGIC TESTS

PRG1=BASIC OUTPUT LOGIC TESTS

PRG2=READER TEST

PRG3=TEST TAPE GENERATOR, PUNCHES CONTENTS OF LOC 00001 AND 00022.

PRG4=TEST TAPE GENERATOR, PUNCHES BINARY COUNT PATTERN TEST TAPE.

PRG5=READER EXERCISER. BINARY COUNT PATTERN.

PRG6=READER EXERCISER. READS TAPE PUNCHED WITH ANY 2 CHARACTERS.

2. REQUIREMENTS

2.1 EQUIPMENT

A. STANDARD PDP-15
B. ASR33 OR ASR35 TELETYPE

2.2 STORAGE

LOCATION 00000 THROUGH 01515 ARE USED.

2.3 PRELIMINARY PROGRAMS

ALL PROGRAMS NECESSARY TO INSURE CORRECT OPERATION OF BASIC PROCESSOR SHOULD HAVE BEEN RUN SUCCESSFULLY.

3. LOADING PROCEDURE

TO LOAD THE PROGRAM PROCEED AS FOLLOWS:

A. LOAD THE OBJECT TAPE IN THE TELETYPE READER, OR IN THE HIGH SPEED READER, IF THE SYSTEM HAS SAID READER.
B. SET ADDRESS SWITCHES TO 17700.
C. SET BANK SWITCH TO "ON" POSITION
D. PRESS "RESET"
E. PRESS "READ IN"
F. THE PROGRAM WILL LOAD AND HALT WITH AC = 777777 IF PROGRAM LOADED CORRECTLY. IF THE PROGRAM HALTS WITH AC = 0, A CHECKSUM ERROR HAS OCCURRED. REPEAT THE LOADING PROCEDURE.

1. USE PROCEDURE

1.1 USE PROCEDURE FOR PRG0

A. INSURE TELETYPE IS ON-LINE.
B. LOAD BINARY COUNT PATTERN TEST TAPE IN READER. IF TAPE IS NOT SPLICED INTO A LOOP, POSITION TAPE OVER PUNCHED SECTION, NOT ON THE LEADER.
C. TURN ON READER.
D. SET CONSOLE REGISTER DISPLAY SWITCH TO AC.
E. SET ADDRESS SWITCHES TO 00200.
F. SET AC SWITCHES TO 000000 (SELECTS PRG0).
G. PRESS I/O RESET; PRESS START
H. PROGRAM HALTS AT LOC 00233 TO PERMIT SETTING OF AC SWITCH OPTIONS. NORMAL OPERATION IS WITH AC SWITCHES SET TO 000000.

THE AC SWITCH OPTIONS FOR THIS PROGRAM ARE:

AC00=1 HALTS PROGRAM AT LOC 00320, AT THE END OF CURRENT ROUTINE. THE NUMBER OF THE ROUTINE COMPLETED IS

DISPLAYED IN AC. TO PROCEED, PRESS CONTINUE.

AC01=1 SELECT ROUTINE WHOSE NUMBER IS SET IN AC12 THROUGH AC17. ROUTINE IS SELECTED UPON THE ROUTINE'S COMPLETION.

AC02=1 LOOP PROGRAM. ENTIRE PROGRAM IS REPEATED.

AC12 THRU AC17 = NUMBER OF ROUTINE TO BE SELECTED. AC01 MUST BE 1.

REFER TO APPROPRIATE SECTION UNDER SECTION 9, "PROGRAM DESCRIPTION," FOR SPECIFIC ROUTINE DESCRIPTIONS.*

I. PRESS CONTINUE.
J. PROGRAM IS EXECUTED AND HALTS AT LOC 00275, PROGRAM END HALT, IF NO OPTIONS ARE SET, OR IF NO ERRORS OCCUR.

NOTE

ERRORS ARE INDICATED BY PROGRAM HALTS. REFER TO SECTION 5.1.1, "NORMAL HALTS," OR SECTION 6.1, "ERROR HALTS AND DESCRIPTIONS,"

4.2 USE PROCEDURE FOR PRG1

A. INSURE TELETYPE IS ON-LINE.
B. INSURE THAT READER IS OFF.
C. INSURE THAT THERE IS SUFFICIENT PAPER IN TELEPRINTER.
D. SET CONSOLE REGISTER DISPLAY SWITCH TO AC.
E. SET ADDRESS SWITCHES TO 00200.
F. SET AC SWITCHES TO 000001. (SELECTS PRG1)
G. PRESS I/O RESET; PRESS START.
H. PROGRAM HALTS AT LOC 00233, FOR SETTING OF AC SWITCH OPTIONS. NORMAL OPERATION IS WITH AC SWITCHES SET TO 000000.

THE AC SWITCH OPTIONS FOR THIS PROGRAM ARE:

AC00=1 HALTS PROGRAM AT LOC 00320, AT END OF CURRENT ROUTINE. NUMBER OF ROUTINE JUST COMPLETED IS DISPLAYED IN AC. TO PROCEED, PRESS CONTINUE.

AC01=1 SELECT ROUTINE WHOSE NUMBER IS SET IN AC12 THROUGH AC17. ROUTINE IS SELECTED UPON COMPLETION OF CURRENT ROUTINE.

AC02=1 LOOP PROGRAM. ENTIRE PROGRAM IS REPEATED.

AC12 THRU AC17 = NUMBER OF ROUTINE TO BE SELECTED. AC01 MUST BE 1. (SEE SECTION 9 FOR DESCRIPTIONS.)

I. PRESS CONTINUE.
J. PROGRAM IS EXECUTED AND HALTS AT LOC 00275, PROGRAM END HALTS, IF NO OPTIONS ARE SET, OR IF NO ERRORS OCCUR. (SEE SECTIONS 5.1.1 AND 6.1 FOR HALTS.)

4.3 USE PROCEDURE FOR PRG2

A. INSURE TELETYPE IS ON-LINE,
B. LOAD BINARY COUNT PATTERN TEST TAPE IN READER. IF TAPE IS NOT SPLICED INTO A LOOP, POSITION TAPE OVER PUNCHED SECTION, NOT ON THE LEADER,
C. TURN ON READER,
D. SET CONSOLE REGISTER DISPLAY SWITCH TO AC,
E. SET ADDRESS SWITCHES TO 00200,
F. SET AC SWITCHES TO 000002 (SELECTS PRG2),
G. PRESS I/O RESET; PRESS START,
H. PROGRAM HALTS AT LOC 00233 FOR SETTING OF AC SWITCH OPTIONS, NORMAL OPERATION IS WITH AC SWITCHES SET TO 000000,

THE AC SWITCH OPTIONS FOR THIS PROGRAM ARE:

AC00=1 HALTS PROGRAM AT LOC 00320, UPON COMPLETION OF CURRENT ROUTINE, NUMBER OF COMPLETED ROUTINE IS DISPLAYED IN AC, TO PROCEED, PRESS CONTINUE,

AC01=1 SELECT ROUTINE WHOSE NUMBER IS SET IN AC12 THROUGH AC17, ROUTINE IS SELECTED UPON ROUTINE'S COMPLETION,

AC02=1 LOOP PROGRAM, ENTIRE PROGRAM IS REPEATED,

AC12 THRU 17 = NUMBER OF ROUTINE TO BE SELECTED, AC01 MUST BE 1. (SEE SECTION 9 FOR DESCRIPTION.)

I. PRESS CONTINUE,
J. PROGRAM IS EXECUTED AND HALTS AT LOC 00275, PROGRAM END HALT, IF NO OPTIONS ARE SET, OR IF NO ERRORS OCCUR, (SEE SECTION 5.1.1 OR SECTION 6.1 FOR HALTS.)

4.4 USE PROCEDURE FOR PRG3

A. INSURE TELETYPE IS ON-LINE AND FULL DUPLEX MODE,
B. TURN OFF TELETYPE READER,
C. LOAD BLANK TAPE IN PUNCH, TURN ON PUNCH,
D. DEPOSIT IN LOC 00021 AND 00022(8), THE 8 BIT CODE FOR CHARACTERS TO BE PUNCHED,
E. SET ADDRESS SWITCHES TO 00200,
F. SET AC SWITCHES TO 000003, (SELECTS PRG3)
G. PRESS I/O RESET, PRESS START,
H. PROGRAM PUNCHES TEST TAPE CONTINUOUSLY UNTIL STOPPED BY USER,

4.5 USE PROCEDURE FOR PRG4

A. INSURE TELETYPE IS ON-LINE.
B. TURN OFF TELETYPE READER.
C. LOAD BLANK TAPE IN PUNCH; TURN ON PUNCH.
D. SET ADDRESS SWITCHES TO 00200.
E. SET AC SWITCHES TO 000004; (SELECTS PRG4)
F. PRESS I/O RESET; PRESS START.
G. PROGRAM PUNCHES BINARY COUNT PATTERN TEST TAPE CONTINUOUSLY UNTIL STOPPED BY USER.

4.6 USE PROCEDURE FOR PRG5

A. INSURE TELETYPE IS ON-LINE.
B. LOAD BINARY COUNT PATTERN TEST TAPE IN READER. IF TAPE IS NOT SPLICED INTO A LOOP, POSITION TAPE OVER PUNCHED SECTION, AND NOT ON THE LEADER.
C. TURN ON READER.
D. SET CONSOLE REGISTER DISPLAY SWITCH TO AC.
E. SET ADDRESS SWITCHES TO 00200.
F. SET AC SWITCHES TO 000005; (SELECTS PRG5)
G. PRESS I/O RESET; PRESS START.
H. PROGRAM STARTS READING TAPE. SET AC SWITCHES TO 040000 TO CAUSE PROGRAM TO HALT-ON-ERROR.
I. PROGRAM RUNS CONTINUOUSLY, UNLESS ERRORS OCCUR, OR AC SWITCH OPTIONS CAUSE A HALT. THE AC SWITCH OPTIONS FOR THIS PROGRAM ARE:

AC00=1 PROGRAM HALTS AT LOC 00652. AC DISPLAYS THE ACCUMULATED ERROR COUNT, IF ANY.

AC03=1 HALT-ON-ERROR. BAD CHARACTER IS DISPLAYED IN AC.

AC04=1 PROGRAM READS TAPE WITH RANDOM DURATION STALLS AFTER EACH CHARACTER.

AC05=1 PROGRAM LOCKS ON CURRENT STALL BEING USED (AC4 MUST BE 1). (SEE SECTION 5.1.1 OR 6.1 FOR HALTS.)

4.7 USE PROCEDURE FOR PRG6

A. INSURE TELETYPE IS ON-LINE;
B. LOAD READER WITH 2 CHARACTER TEST TAPE,
C. TURN ON READER,
D. DEPOSIT IN LOC 00021 AND 00022 THE 8-BIT CODES FOR THE CHARACTERS PUNCHED IN THE TEST TAPE,
E. SET CONSOLE REGISTER DISPLAY SWITCH TO AC,
F. SET ADDRESS SWITCHES TO 00200,
G. SET AC SWITCHES TO 000006, (SELECTS PRG6)
H. PRESS I/O RESET; PRESS START,
I. PROGRAM STARTS READING TAPE, SET AC SWITCHES TO 040000 TO CAUSE PROGRAM TO HALT-ON-ERROR,
J. PROGRAM RUNS CONTINUOUSLY, UNLESS ERRORS OCCUR, OR IF CAUSED TO HALT BY AC SWITCH OPTIONS,

THE AC SWITCH OPTIONS FOR THIS PROGRAM ARE;

AC00=1 PROGRAM HALTS AT LOC 00652, AC DISPLAYS THE ACCUMULATED ERROR COUNT, IF ANY,

AC03=1 HALT-ON-ERROR, BAD CHARACTER IS DISPLAYED IN AC,

AC04=1 PROGRAM READS TAPE WITH RANDOM DURATION STALLS AFTER EACH CHARACTER,

AC05=1 PROGRAM LOCKS ON CURRENT STALL BEING USED (AC04 MUST BE 1), (SEE SECTIONS 5,1,1 OR 6,1 FOR HALTS,)

5. OPERATING PROCEDURE

5.1 PROGRAM AND/OR OPERATOR ACTION

5.1.1 NORMAL HALTS

LOC 00033 — OPTION SET HALT, OCCURS DURING PRG0, PRG1, AND PRG2 TO PERMIT SETTING OF OPTIONS. SET OPTIONS DESIRED AND PRESS CONTINUE.

LOC 00275 — PROGRAM END HALT, OCCURS AT END OF PRG0, PRG1, OR PRG2 IF NO "LOOP PROGRAM" OPTION IS SET. SET DESIRED OPTIONS AND PRESS CONTINUE. IF NO OPTIONS ARE SET, THIS HALT REOCCURS.

LOC 00320 — ROUTINE END HALT, OCCURS AT END OF CURRENT ROUTINE DURING PRG0, PRG1, OR PRG2 EXECUTION IF AC$0 IS ON. AC DISPLAYS ROUTINE NUMBER, TO PROCEED, PRESS CONTINUE.

LOC 00652 — HALT, OCCURS DURING PRG5 AND PRG6 IF AC$0 IS ON. THE ACCUMULATED ERROR COUNT, IF ANY, IS DISPLAYED BY THE AC. TO PROCEED, PRESS CONTINUE.

6. ERRORS

6.1 ERROR HALTS AND DESCRIPTION

LOC 00177 — INCORRECT PROGRAM NUMBER SELECTED. SET THE CORRECT PROGRAM NUMBER IN ACS14 THROUGH ACS17 AND PRESS CONTINUE.

LOC 00256 — NON-EXISTENT ROUTINE NUMBERS SELECTED. SET THE CORRECT ROUTINE NUMBER IN ACS12 THROUGH ACS17 AND PRESS CONTINUE.

LOC 00542 — SYNC ERROR. OCCURS DURING PRG2 AND PRG5 WHEN THE "SYNC" SUBROUTINE DOES NOT FIND AN ALL 1S CHARACTER WITHIN 256 CHARACTERS. TO RETRY, PRESS CONTINUE.

LOC 00643 — PRG5, PRG6 READ ERROR. OCCURS IF READ ERROR IS DETECTED AND ACS3 IS ON. BAD CHARACTER IS DISPLAYED IN AC. PRESS CONTINUE.

LOC 00645 — FOLLOW UP HALT TO PRG5 AND PRG6 READ ERROR. EXPECTED CHARACTER IS DISPLAYED IN AC.

LOC 00674 — ALIGN ERROR HALT. OCCURS DURING PRG6 IF THE "ALIGN" SUBROUTINE CANNOT MATCH THE DATA READ FROM PAPER TAPE WITH DATA STORED IN LOC 00021 AND 00022(8). CHECK THAT CORE DATA IS CORRECT AND THAT THE TEST TAPE IS CORRECT AND IN OPERATING CONDITION. PRESS CONTINUE TO RETRY.

LOC 00741 — PRG0, ROUTINE 0 ERROR HALT. I/O COMMAND 700311 FAILED TO CLEAR THE AC. PRESSING CONTINUE ENTERS SCOPE LOOP THAT SETS THE AC TO ALL 1S, ISSUES 700311 COMMAND TO CLEAR THE AC AND REPEATS. PROGRAM MUST BE RESTARTED MANUALLY.

LOC 00765 — PRG0, ROUTINE 1 ERROR HALT. AT LEAST 200 MS AFTER 700322 COMMAND THE READER FLAG FAILED TO SET, OR KSF INSTRUCTION IS FAILING. PRESSING CONTINUE REPEATS THE TEST.

LOC 01010 — PRG0, ROUTINE 2 ERROR HALT A. SAME AS PRG0, ROUTINE 1 ERROR HALT.

LOC 01012 — PRG0, ROUTINE 2 ERROR HALT B. KSF FAILURE, WITH READER FLAG SET, KSF INSTRUCTION FAILED TO SKIP. PRESSING CONTINUE ENTERS SCOPE LOOP THAT SKIPS ON FLAG CONTINUOUSLY, MANUAL RESTART.

(6.1 CONT'D)

LOC 01041 PRG0, ROUTINE 3 ERROR HALT A. SAME AS PRG0, ROUTINE 1 ERROR HALT.

LOC 01043 PRG0, ROUTINE 3 ERROR HALT B. THE 700312 COMMAND (KRB) FAILED TO RESET THE READER FLAG OR THE KSF COMMAND SKIPPED WITH THE READER FLAG = 0. PRESSING CONTINUE ENTERS SCOPE LOOP THAT CLEARS THE FLAG AND SKIPS ON FLAG CONTINUOUSLY. MANUAL RESTART.

LOC 01063 PRG0, ROUTINE 4 ERROR HALT A. UNEXPECTED INTERRUPT. AC CONTAINS THE IORS WORD. TURN OFF DEVICE CAUSING THE INTERRUPT (OTHER THAN THE TELETYPE). PRESS CONTINUE TO REPEAT TEST.

LOC 01100 PRG0, ROUTINE 4 ERROR HALT B. WITH READER FLAG = 1, AND INTERRUPT ENABLED, NO INTERRUPT OCCURRED. PRESSING CONTINUE ENTERS SCOPE LOOP THAT TURNS ON INTERRUPT CONTINUOUSLY. MANUAL RESTART.

LOC 01131 PRG0, ROUTINE 5 ERROR HALT. TIMING ERROR. FLAG NOT 1 110MS AFTER 700322 COMMAND. PRESSING CONTINUE ENTERS SCOPE LOOP THAT READS TAPE CONTINUOUSLY TO AID IN TIMING ADJUSTMENT.

LOC 01171 PRG0, ROUTINE 6 ERROR HALT A, REREAD ERROR. A REREAD OF THE TELETYPE READER BUFFER DID NOT MATCH WITH THE ORIGINAL READ. THE "NEW" CHARACTER IS DISPLAYED IN AC. PRESS CONTINUE.

LOC 01173 PRG0, ROUTINE 6 ERROR HALT B. FOLLOW UP HALT TO PRG0, ROUTINE 6 ERROR HALT A. THE "OLD" CHARACTER IS DISPLAYED IN AC. PRESSING CONTINUE RESUMED TEST.

(6.1 CONT'D)

LOC 01227	PRG1, ROUTINE 0 ERROR HALT A, AT LEAST 200 MS AFTER TLS COMMAND THE PRINTER FLAG FAILED TO SET OR TSF COMMAND FAILED TO SKIP, PRESSING CONTINUE RETRIES THE TEST.
LOC 01231	PRG1, ROUTINE 0 ERROR HALT B, WITH FLAG = 1, TSF COMMAND FAILED TO SKIP, PRESSING CONTINUE ENTERS SCOPE LOOP THAT SKIPS ON FLAG CONTINUOUSLY, MANUAL RESTART.
LOC 01251	PRG1, ROUTINE 1 ERROR HALT, TCF COMMAND FAILED TO CLEAR THE FLAG, OR TSF COMMAND SKIPPED WITH FLAG = 0, PRESSING CONTINUE ENTERS SCOPE LOOP THAT CLEARS THE FLAG AND THEN SKIPS ON FLAG CONTINUOUSLY, MANUAL RESTART.
LOC 01274	PRG1, ROUTINE 2 ERROR HALT, TCF COMMAND FAILED TO CLEAR THE FLAG, PRESSING CONTINUE ENTERS SCOPE LOOP THAT ISSUES TCF COMMAND CONTINUOUSLY, MANUAL RESTART.
LOC 01317	PRG1, ROUTINE 3 ERROR HALT A, UNEXPECTED INTERRUPT, TURN OFF ANY DEVICE THAT MAY BE CAUSING THE INTERRUPT, (THE TELETYPE READER MUST BE OFF,) PRESS CONTINUE TO REPEAT TEST.
LOC 01335	PRG1, ROUTINE 3 ERROR HALT B, WITH PRINTER/PUNCH FLAG = 1 AND INTERRUPT ENABLED, NO INTERRUPT OCCURRED, PRESSING CONTINUE ENTERS SCOPE LOOP THAT TURNS ON INTERRUPT CONTINUOUSLY, MANUAL RESTART.
LOC 01366	PRG1, ROUTINE 4 ERROR HALT, TIMING ERROR, FLAG NOT 1 110MS AFTER TLS COMMAND, PRESSING CONTINUE ENTERS SCOPE LOOP THAT RUNS PRINTER CONTINUOUSLY TO AID IN TIMING ADJUSTMENT, MANUAL RESTART.

(6.1 CONT'D)

LOC 01422	PRG2, ROUTINE 0 ERROR HALT A, READ ERROR, BAD CHARACTER IN AC, PRESS CONTINUE,
LOC 01424	PRG2, ROUTINE 0 ERROR HALT B, FOLLOW UP HALT TO PRG2, ROUTINE 0 ERROR HALT A, EXPECTED CHARACTER IS DISPLAYED IN AC, PRESSING CONTINUE RESUMES TESTING,
LOC 01453	PRG2, ROUTINE 1 ERROR HALT A, READ ERROR, BAD CHARACTER IN AC, PRESS CONTINUE,
LOC 01455	PRG2, ROUTINE 1 ERROR HALT B, FOLLOW UP HALT TO PRG2, ROUTINE 1 ERROR HALT A, EXPECTED CHARACTER IS DISPLAYED IN AC, PRESS CONTINUE TO RESUME TESTING,
LOC 01510	PRG2, ROUTINE 2 ERROR HALT A, READ ERROR, BAD CHARACTER IN AC, PRESS CONTINUE,
LOC 01512	PRG2, ROUTINE 2 ERROR HALT B, FOLLOW UP HALT TO PRG2, ROUTINE 2 ERROR HALT A, EXPECTED CHARACTER IS DISPLAYED IN AC, PRESS CONTINUE TO RESUME TESTING,

7. RESTRICTIONS

7.1 STARTING RESTRICTIONS

ALL PROGRAMS MUST BE STARTED AT ADDRESS 00200.

7.2 OPERATING RESTRICTIONS

PRG0 AND PRG1 MUST HAVE BEEN RUN SUCCESSFULLY PRIOR TO EXECUTING ANY OTHER PROGRAM.

PROBLEMS DETECTED DURING EXECUTION OF PRG0 AND PRG1 SHOULD BE CORRECTED AS THEY OCCUR.

8. MISCELLANEOUS

8.1 EXECUTION TIME

```
PRG0 EXECUTION TIME:    1 MIN  15 SEC (MAX)
PRG1 EXECUTION TIME:    30 SEC (MAX)
PRG2 EXECUTION TIME:    20 MIN (MAX)
PRG3 THROUGH PRG6 ARE CONTINUOUSLY RUN PROGRAMS.
```

8.2 TEST TAPES

MAINDEC-00-D2G3-PT BINARY COUNT PATTERN TEST TAPE IS PROVIDED WITH THIS PROGRAM. FOR EASY USE, THE TAPE SHOULD BE SPLICED INTO A LOOP, MATCHING THE PATTERN OF THE SPLICE POINT.

9. PROGRAM DESCRIPTION

THE PDP-15, ASR33/35 TELETYPE TESTS PART 1, CONSISTS OF 7 PROGRAMS NUMBERED FROM 0 TO 6.

9.1 PRG0 - BASIC INPUT LOGIC TESTS

THIS PROGRAM CONTAINS 7 ROUTINES NUMBERED FROM 0 TO 6 (OCTAL).

RTN0 CHECKS THAT IOT 700311 IS ABLE TO CLEAR THE AC, DONE 1000 TIMES.

RTN1 CHECKS THAT READER FLAG IS SET 200 MS AFTER 700322 IOT (SELECT READER). FAILURE TO SKIP INDICATES THAT FLAG IS NOT SET, OR KSF COMMAND FAILURE TO SKIP. DONE 100 TIMES.

RTN2 CHECKS THAT KSF COMMAND (IOT 700301) SKIPS WITH FLAG SET, DONE 1000 TIMES.

RTN3 CHECKS THAT KSF COMMAND (IOT 700301) DOES NOT SKIP WITH FLAG = 0. DONE 500 TIMES.

RTN4 CHECKS THAT NO OTHER DEVICE CAN CAUSE AN INTERRUPT AND THAT READER IS CAPABLE OF INTERRUPTING.

RTN5 READER TIMING TEST.

RTN6 READS A CHARACTER FROM TAPE AND SAVES IT. IT THEN REREADS THE TTI STATICALLY 1000 TIMES TO CHECK FOR CONSISTENT READING FROM TTI. 256 CHARACTERS ARE READ IN THIS MANNER.

9.2 PRG1 - BASIC OUTPUT LOGIC TESTS

THIS PROGRAM CONTAINS 5 ROUTINES NUMBERED FROM 0 TO 4.

RTN0 CHECKS THAT PRINTER/PUNCH FLAG IS SET 200 MS AFTER TLS COMMAND. FAILURE TO SKIP IN FLAG INDICATES THAT FLAG IS NOT SET OR TSF FAILURE. IF TSF SKIPS, ROUTINE THEN SKIPS ON FLAG 1000 TIMES TO CHECK FOR RELIABLE SKIPPING.

RTN1 CHECKS THAT TSF COMMAND DOES NOT SKIP WITH FLAG = 0, DONE 1000 TIMES.

RTN2 CHECKS THAT TCF COMMAND IS ABLE TO CLEAR FLAG, DONE 100 TIMES.

RTN3 CHECKS THAT NO OTHER DEVICE CAN INTERRUPT AND THAT PRINTER/PUNCH IS ABLE TO INTERRUPT.

RTN4 TIMING TEST.

9.3 PRG2 - READER TEST

THIS PROGRAM CONTAINS 3 ROUTINES NUMBERED FROM 0 TO 2.

RTN0 READS 4095 CHARACTERS OF BINARY COUNT PATTERN AT FULL SPEED.

RTN1 READS 2000 CHARACTERS OF BINARY COUNT PATTERN WITH RANDOM STALLS BETWEEN CHARACTERS.

RTN2 READS 100 RANDOM LENGTH CHARACTER BLOCKS. FIXED STALL BETWEEN CHARACTERS IN A BLOCK. STALL IS CHANGED FOR EACH BLOCK, AND IS DETERMINED AT RANDOM.

9.4 PRG3 - TEST TAPE GENERATOR

THIS PROGRAM PUNCHES TEST TAPE WITH CHARACTERS WHOSE CODE IS SET IN LOC 00021 AND 00022.

9.5 PRG4 - TEST TAPE GENERATOR

PUNCHES BINARY COUNT PATTERN TEST TAPE.

9.6 PRG5 - READER EXERCISER, BINARY COUNT PATTERN

THIS PROGRAM READS AND CHECKS BINARY COUNT PATTERN TEST TAPE, NORMAL TEST MODE IS WITH FULL SPEED OPERATION, ACS4 = 1 GIVES RANDOM STALLS BETWEEN CHARACTERS, ACS5 = 1 LOCKS PROGRAM ON CURRENT STALL (ACS4 MUST BE ON),

9.7 PRG6 - READER EXERCISER, LOC 00021 AND 00022

READS TEST TAPE PUNCHED WITH ANY 2 TEST CHARACTERS, LOC 00021 AND 00022 MUST CONTAIN THE CODES FOR CHARACTERS IN TEST TAPE, NORMAL TEST MODE IS WITH FULL SPEED OPERATION, ACS4 = 1 GIVES RANDOM STALLS BETWEEN CHARACTERS; ACS5 = 1 LOCKS PROGRAM IN CURRENT STALL, (ACS4 MUST BE ON),

(LAST PAGE)

```
/PDP-15 AND ASR33/35 TELETYPE TEST - PART1
/COPYRIGHT 1969, DIGITAL EQUIPMENT CORP., MAYNARD, MASS.
/
/PRGØ-BASIC INPUT CONTROL LOGIC TEST. USES READER
/PRG1-BASIC OUTPUT CONTROL LOGIC TEST. USES PRINTER
/PRG2-READER TEST
/PRG3-TEST TAPE GENERATOR. PUNCHES CONTENTS OF LOC ØØ21 AND ØØ22
/PRG4-PUNCHES BINARY COUNT PATTERN TEST TAPE.
/PRG5-READER EXERCISER. BINARY COUNT PATTERN
/PRG6-READER EXERCISER. READS TAPE PUNCHED WITH ANY
/         2 CHARACTERS. LOC ØØ21 AND ØØ22 CONTAINS CODES
/         FOR CHARACTERS IN TAPE.
/
/ACS-OPTIONS
/
/ACSØ-HALT AT END OF ROUTINE. RTN NUMBER IN AC FOR PRGØ,PRG1,PRG2.
/         -HALT. ERROR COUNT IN AC FOR PRG5 AND PRG6.
/ACS1-SELECT ROUTINE WHOSE NUMBER IS SET IN ACS12 THROUGH ACS17.
/         (PRGØ,PRG1,PRG2).
/ACS2-LOOP PROGRAM. (PRGØ,PRG1,PRG2)
/ACS3-HALT ON ERROR. BAD CHARACTER IN AC (PRG5,PRG6)
/ACS4-STALL (PRG5,PRG6)
/ACS5-LOCK ON STALL (PRG5,PRG6)
          .EJECT
```

```
                                .ABS
00020                           .LOC 20
00020   000000      KSTART      0
00021   000000      PTEMP       0
00022   000000      PTEMP1      0
00023   000000      DELAYM      0
00024   620002                  JMP* 2
00025   000000                  0
00026   000077      TSTMSK      77
00027   000017      PRGMSK      17
00030   777772      PRGLIM      -6
00031   000000      PRGNUM      0
00032   000033      PSW         PRGTAB
00033   000717      PRGTAB      PRG0
00034   001201                  PRG1
00035   001373                  PRG2
00036   000571                  PRG3
00037   000576                  PRG4
00040   000602                  PRG5
00041   000611                  PRG6
00042   000000      TEMP        0
00043   000000      CURTST      0
00044   000000      RTNNO       0
00045   000000      NXTST       0
00046   000000      MSCTR       0
00047   000000      MILCTR      0
00050   777142      MIL1        -636
00051   000000      CTRA        0
00052   000000      CTRB        0
00053   000000      SCNT        0
00054   777401      MRBOUT      -377
00055   000000      PFLAG       0
00056   040000      SR3MSK      040000
00057   020000      SR4MSK      020000
00060   010000      SR5MSK      010000
00061   000001      K1          1
        700322      KRA=700322
                    /
                                .EJECT
```

```
00177                           .LOC 177
00177   740040                  HLT                             /INCORRECT PROGRAM NUMBER
00200   750004  START           LAS                             /READ SWITCHES
00201   500027                  AND     PRGMSK                  /AND WITH PRG MASK
00202   340030                  TAD     PRGLIM                  /ADD PROGRAM LIMIT
00203   740300                  SMA!SZA                         /VALID PROGRAM NUMBER?
00204   600177                  JMP     177                     /NO.
00205   750004                  LAS                             /YES, READ SWITCHES
00206   500027                  AND     PRGMSK
00207   040031                  DAC     PRGNUM                  /SAVE PROGRAM NUMBER
00210   340032                  TAD     PSW                     /DEVELOPE PROGRAM
00211   040042                  DAC     TEMP                    /ADDRESS.
00212   220042                  LAC*    TEMP
00213   040221                  DAC     PRGADR                  /STORE ADDRESS
00214   100545                  JMS     MOVE                    /INITIALIZE INTERRUPT
00215   000024                  24                              /AREA
00216   000001                  1
00217   777776                  -2
00220   620221                  JMP*    .+1
00221   000000  PRGADR          0
00233                           .LOC 233
00233   750040  SRSET           HLT!CLA                         /SELECT OPTIONS
00234   200020  GETRDY          LAC     KSTART                  /GET ADDR OF 1ST RTN
00235   040045                  DAC     NXTST                   /STORE AT NXTST
00236   100277                  JMS     FORWD
00237   750004                  LAS                             /READ SWITCHES
00240   742010                  RTL
00241   740400                  SNL                             /SELECT ROUTINE?
00242   620043                  JMP*    CURTST                  /NO, START WITH CURRENT ROUTINE
00243   750004                  LAS                             /YES, READ SWITCHES
00244   500026                  AND     TSTMSK
00245   740001                  CMA
00246   340061                  TAD     K1
00247   340044                  TAD     RTNNO
00250   751200                  SNA!CLA                         /IS IT THIS RTN?
00251   620043                  JMP*    CURTST                  /YES, GO DO IT.
00252   200045                  LAC     NXTST                   /NO.
00253   340061                  TAD     K1                      /LAST ROUTINE?
00254   750200                  SZA!CLA
00255   600236                  JMP     GETRDY+2  /NO.
00256   750040  INCRTN          HLT!CLA                         /YES, INCORRECT ROUTINE NUMBER.
00257   600234                  JMP     GETRDY
                                .EJECT
```

```
00260   100313      CHAIN   JMS     SHALT           /HALT? (SR0)
00261   750004              LAS                     /READ SWITCHES
00262   742010              RTL
00263   741400              SZL                     /ROUTINE SELECT? (SR1)
00264   600234              JMP     GETRDY          /YES.
00265   200045              LAC     NXTST
00266   340061              TAD     K1
00267   750200              SZA!CLA                 /LAST RTN?
00270   600236              JMP     GETRDY+2  /NO.
00271   750004              LAS                     /YES.
00272   742010              RTL
00273   751100              SPA!CLA                 /LOOP PROGRAM? (SR2)
00274   600234              JMP     GETRDY
00275   750040      PRGEND  HLT!CLA                 /PROG END HALT
00276   600260              JMP     CHAIN
00277   000000      FORWD   0
00300   220045              LAC*    NXTST           /GET NEXT RTN NO.
00301   040044              DAC     RTNNO           /STORE AT RTNNO
00302   440045              ISZ     NXTST
00303   200045              LAC     NXTST           /SET CURRENT
00304   040042              DAC     TEMP            /RTN NUMBER
00305   440045              ISZ     NXTST
00306   200045              LAC     NXTST           /SET CURRENT
00307   040043              DAC     CURTST          /ROUTINE ADDR
00310   220042              LAC*    TEMP            /SET NEXT
00311   040045              DAC     NXTST           /RTN ADDR
00312   620277              JMP*    FORWD           /EXIT
                    /
00313   000000      SHALT   0
00314   750004              LAS                     /READ SW
00315   740100              SMA                     /HALT? (SR0)
00316   620313              JMP*    SHALT           /NO, EXIT
00317   200044              LAC     RTNNO
00320   740040              HLT                     /ROUTINE END HALT.
00321   620313              JMP*    SHALT
00322   000000      STCTR   0
00323   220322              LAC*    STCTR           /GET LOC ADDRESS
00324   040042              DAC     TEMP            /SAVE AT TEMP
00325   440322              ISZ     STCTR
00326   220322              LAC*    STCTR           /GET COUNT AND
00327   060042              DAC*    TEMP            /STORE AT DESIRED LOC.
00330   440322              ISZ     STCTR
00331   754000              CLA!CLL
00332   620322              JMP*    STCTR           /EXIT
        100322      SETLOC=JMS STCTR
                    .EJECT
```

```
00333  000000  DLYMS   0
00334  200023          LAC     DELAYM          /GET MS COUNT
00335  040046          DAC     MSCTR           /STORE IN MSCTR
00336  200050          LAC     MIL1            /GET 1 MS CONSTANT
00337  040047          DAC     MILCTR          /STORE AT MILCTR
00340  440047          ISZ     MILCTR          /DELAYED 1 MS?
00341  600340          JMP     .-1             /NO
00342  440046          ISZ     MSCTR           /YES. DONE DELAYING?
00343  600336          JMP     .-5             /NO.
00344  754000          CLA!CLL                 /YES.
00345  620333          JMP*    DLYMS
                /RANDOM NUMBER GENERATOR
00346  000000  RANGEN  0
00347  200374          LAC     RANDEX
00350  541514          SAD     (RANTBL+10
00351  741000          SKP
00352  600362          JMP     RANTAD-1
00353  201515          LAC     (RANTBL
00354  040374          DAC     RANDEX
00355  200373          LAC     RANCON
00356  745100          SPA!CLL
00357  744002          STL
00360  740010          RAL
00361  040373          DAC     RANCON
00362  220374          LAC*    RANDEX
00363  340373  RANTAD  TAD     RANCON
00364  060374          DAC*    RANDEX
00365  200405          LAC     RANSAV
00366  740020          RAR
00367  360374          TAD*    RANDEX
00370  040405          DAC     RANSAV
00371  440374          ISZ     RANDEX
00372  620346          JMP*    RANGEN
                /
00373  123456  RANCON  123456
00374  000405  RANDEX  RANTBL+10
00375  654321  RANTBL  654321
00376  361416          361416
00377  055363          055363
00400  546060          546060
00401  243035          243035
00402  762572          762572
00403  453237          453237
00404  150214          150214
00405  000000  RANSAV  0
                /
                        .EJECT
```

```
00406   000000   INITPT   0                          /INITIALIZE COUNT PATTERN
00407   140412            DZM      PT0              /0 TO PT0
00410   750000            CLA                       /0 TO AC
00411   620406            JMP*     INITPT           /EXIT.
00412   000000   PT0      0
00413   000000   PT1      0
00414   000377   PTMSK    377
00415   000000   GETPTT   0
00416   200412            LAC      PT0              /PT0 TO PT1
00417   040413            DAC      PT1
00420   340061            TAD      K1
00421   500414            AND      PTMSK
00422   040412            DAC      PT0              /PT0+1 TO PT0
00423   200413            LAC      PT1              /GET PT1
00424   620415            JMP*     GETPTT           /EXIT
                 /
00425   000000   PUNCH    0
00426   440055            ISZ      PFLAG            /SET PFLAG
00427   700406            TLS                       /PUNCH/PRINT
00430   200055            LAC      PFLAG
00431   750200            SZA!CLA                   /FLAG RESET?
00432   741000            SKP                       /NO
00433   600436            JMP      .+3
00434   700401            TSF                       /SAVE PRINTING?
00435   600430            JMP      .-5              /NO.
00436   700402            TCF                       /YES, CLEAR FLAG
00437   140055            DZM      PFLAG            /0 TO PFLAG
00440   620425            JMP*     PUNCH            /EXIT.
                 /
00441   000000   STALL    0
00442   750004            LAS                       /READ SWITCHES
00443   500057            AND      SR4MSK
00444   751200            SNA!CLA                   /STALL? (ASC4)
00445   620441            JMP*     STALL            /NO, EXIT
00446   750004            LAS                       /YES, READ SWITCHES
00447   500060            AND      SR5MSK
00450   750200            SZA!CLA                   /LOCK ON STALL?
00451   741000            SKP                       /YES.
00452   100474            JMS      DLCNT            /GENERATE RANDOM COUNT
00453   100333            JMS      DLYMS            /STALL.
00454   620441            JMP*     STALL            /EXIT.
                          .EJECT
```

```
                        /
00455   000000  CHRCNT  0
00456   100346          JMS     RANGEN          /SET RANDOM NUMBER.
00457   500473          AND     CRMSK           /REMOVE EXCESS BITS
00460   741200          SNA                     /ZERO?
00461   600456          JMP     CHRCNT+1  /YES.
00462   740001          CMA
00463   340061          TAD     K1
00464   040053          DAC     SCNT
00465   220455          LAC*    CHRCNT
00466   040042          DAC     TEMP
00467   200053          LAC     SCNT
00470   060042          DAC*    TEMP            /STORE AT DESIRED ADDR
00471   440455          ISZ     CHRCNT
00472   620455          JMP*    CHRCNT          /EXIT
00473   000077  CRMSK    77
00474   000000  DLCNT    0
00475   100346          JMS     RANGEN          /GET RANDOM NUMBER
00476   500505          AND     DLYMSK          /REMOVE EXCESS BITS
00477   741200          SNA                     /ZERO?
00500   600475          JMP     DLCNT+1         /YES
00501   740001          CMA
00502   340061          TAD     K1
00503   040023          DAC     DELAYM
00504   620474          JMP*    DLCNT           /EXIT
00505   000277  DLYMSK   277
                        /
00506   000000  CHCK     0
00507   040523          DAC     WCHK            /AC TO WCHK
00510   220506          LAC*    CHCK            /GET COMPARE DATA
00511   740001          CMA
00512   340061          TAD     K1
00513   340523          TAD     WCHK
00514   440506          ISZ     CHCK            /SET UP FOR UNEG EXIT
00515   750200          SZA!CLA                 /EQUAL?
00516   600521          JMP     .+3             /NO.
00517   440506          ISZ     CHCK            /YES, SET UP EQUAL EXIT
00520   620506          JMP*    CHCK            /EXIT
00521   200523          LAC     WCHK            /RESTORE AC
00522   620506          JMP*    CHCK            /EXIT
00523   000000  WCHK     0
                        /
                        .EJECT
```

```
                        /
00524   000000  SYNK    0
00525   100322          SETLOC                  /-256 TO CTSK
00526   000544          CTSK
00527   777400          -400
00530   700322  SYNKA   KRA                     /START READER.
00531   700301          KSF                     /FLAG1?
00532   600531          JMP     .-1             /NO
00533   700312          KRB                     /READ
00534   340054          TAD     MRBOUT          /
00535   750200          SZA!CLA                 /377?
00536   741000          SKP                     /NO.
00537   620524          JMP*    SYNK            /YES, EXIT
00540   440544          ISZ     CTSK
00541   600530          JMP     SYNKA
00542   740040          HLT                     /CAN'T SYNK
00543   600525          JMP     SYNK+1          /REPEAT.
00544   000000  CTSK    0
00545   000000  MOVE    0                       /MOVE SUBROUTINE
00546   220545          LAC*    MOVE            /GET AND STORE
00547   040566          DAC     FADDR           /"FROM" ADDR
00550   440545          ISZ     MOVE
00551   220545          LAC*    MOVE            /GET AND STORE
00552   040567          DAC     TADDR           /"TO" ADDR
00553   440545          ISZ     MOVE
00554   220545          LAC*    MOVE            /GET AND STORE
00555   040570          DAC     MCTR            /MOVE COUNT.
00556   440545          ISZ     MOVE
00557   220566  MOVEA   LAC*    FADDR           /GET "NEW WORD"
00560   060567          DAC*    TADDR           /STORE AT "TO" LOC.
00561   440566          ISZ     FADDR           /UPDATE ADDRESSES
00562   440567          ISZ     TADDR
00563   440570          ISZ     MCTR            /DONE MOVING?
00564   600557          JMP     MOVEA           /NO.
00565   620545          JMP*    MOVE            /YES, EXIT.
00566   000000  FADDR   0
00567   000000  TADDR   0
00570   000000  MCTR    0
                        /
                        .EJECT
```

```
                        /PROGRAM 3 PUNCHES TEST TAPE WITH CHARACTERS STORED IN
                        /SYMBOLIC LOCATION PTEMP AND PTEMP1.
00571   200021  PRG3    LAC     PTEMP           /GET C(PTEMP)
00572   100425          JMS     PUNCH           /PUNCH
00573   200022          LAC     PTEMP1          /GET C(PTEMP1)
00574   100425          JMS     PUNCH           /PUNCH
00575   600571          JMP     PRG3            /REPEAT.
                        /PROGRAM4 PUNCHES TEST TAPE WITH BINARY COUNT PATTERN
00576   100406  PRG4    JMS     INITPT          /INITIALIZE PATTERN
00577   100415          JMS     GETPTT          /GET BINARY CHARACTER
00600   100425          JMS     PUNCH           /PUNCH IT
00601   600577          JMP     .-2             /REPEAT.
                        /PROGRAM 5 READS COUNT PATTERN. STALLS OPTIONAL
00602   100524  PRG5    JMS     SYNK            /SYNK TAPE
00603   140654          DZM     ERRCTR          /0 TO ERRCTR
00604   100406          JMS     INITPT          /INITIALIZE BINARY PATTERN
00605   700322          KRA                     /START READER
00606   100415  SRT0A   JMS     GETPTT          /GET BINARY NUMBER
00607   100617          JMS     READCK          /READ AND CHECK CHARACTER
00610   600606          JMP     SRT0A           /REPEAT
                        /PROGRAM 6 READS DATA FROM TAPE AND MATCHES AGAINST CHARACTERS
                        /IN PTEMP AND PTEMP1. RANDOM DELAY BETWEEN CHARACTERS OPTIONAL
00611   140654  PRG6    DZM     ERRCTR          /0 TO ERRCTR
00612   100655          JMS     ALIGN
00613   700322          KRA                     /START READER
00614   100705  SRT2A   JMS     GIVE            /GET CHARACTER
00615   100617          JMS     READCK          /READ AND CHECK
00616   600614          JMP     SRT2A           /REPEAT
00617   000000  READCK  0
00620   040627          DAC     SBSP
00621   100441          JMS     STALL
00622   700301          KSF                     /READER READY?
00623   600622          JMP     .-1             /NO, REPEAT.
00624   700312          KRB                     /READ, START READER
00625   700322          KRA
00626   100506          JMS     CHCK
00627   000000  SBSP    0
00630   600632          JMP     ERRCNT          /ERROR.
00631   600646          JMP     HLTTST          /OK.
00632   440654  ERRCNT  ISZ     ERRCTR          /+1 TO ERRCTR
00633   600636          JMP     .+3
00634   750001          CLA!CMA
00635   040654          DAC     ERRCTR          /7777 TO ERRCTR
00636   750004          LAS                     /READ SWITCHES
00637   500056          AND     SR3MSK
00640   751200          SNA!CLA                 /HALT ON ERROR (SR3)
00641   600646          JMP     HLTTST          /NO
00642   200523          LAC     WCHK            /GET BAD CHAR
00643   740040          HLT                     /YES
00644   200627          LAC     SBSP            /GET GOOD CHAR
00645   740040          HLT                     /GOOD CHAR IN AC
                        .EJECT
```

```
00646   750004      HLTTST  LAS
00647   750100              SMA!CLA                 /HALT?
00650   620617              JMP*    READCK          /NO, EXIT
00651   200654              LAC     ERRCTR          /GET ERROR COUNT
00652   740040              HLT                     /ERROR COUNT IN AC
00653   620617              JMP*    READCK          /EXIT
00654   000000      ERRCTR  0
                    /
00655   000000      ALIGN   0
00656   700322              KRA
00657   700301              KSF                     /READER READY?
00660   600657              JMP     .-1
00661   700312              KRB                     /READ CHARACTER
00662   740001              CMA
00663   340061              TAD     K1
00664   040704              DAC     ATEMP
00665   340021              TAD     PTEMP
00666   751200              SNA!CLA                 /IS IT CHAR IN PTEMP?
00667   600676              JMP     AL1             /YES.
00670   200704              LAC     ATEMP           /NO.
00671   340022              TAD     PTEMP1
00672   751200              SNA!CLA                 /IS IT CHAR IN PTEMP1?
00673   600701              JMP     AL2             /YES.
00674   750040              HLT!CLA                 /NO, ERROR.
00675   600656              JMP     ALIGN+1         /REPEAT.
00676   740001      AL1     CMA
00677   040703              DAC     IND             /SET IND TO -1
00700   620655              JMP*    ALIGN
00701   140703      AL2     DZM     IND             /0 TO IND
00702   620655              JMP*    ALIGN
00703   000000      IND     0
00704   000000      ATEMP   0
                    /
00705   000000      GIVE    0
00706   440703              ISZ     IND             /IND=1?
00707   600713              JMP     .+4             /NO
00710   140703              DZM     IND             /YES
00711   200022              LAC     PTEMP1          /GET TEMP1
00712   620705              JMP*    GIVE            /EXIT
00713   750001              CLA!CMA
00714   040703              DAC     IND             /-1 TO IND
00715   200021              LAC     PTEMP           /GET PTEMP
00716   620705              JMP*    GIVE            /EXIT.
                    /
                            .EJECT
```

```
                        /PROGRAM 0. ASR33/35 TELETYPE BASIC INPUT TESTS.
                        /PROGRAM CHECKS INPUT IOT'S, INTERRUPT, AND READER TIMING.
00717   100322  PRG0    SETLOC                          /SET KSTART TO INITIAL
00720   000020          KSTART                          /ROUTINE ADDRESS
00721   000723          P0TS0
00722   600233          JMP     SRSET                   /GO START PROGRAM.
                        /
00723   000000  P0TS0   0
00724   000746          P0TS1
                        /ISSUE 700311 OP WITH AC=777777. AC SHOULD GO TO 0
00725   100322          SETLOC                          /-1000 TO CTRA
00726   000051          CTRA
00727   776030          -1750
00730   750001  P0TS0A  CLA!CMA                         /-1 TO AC
00731   700311          700311                          /CLEAR AC.
00732   740000          NOP
00733   740000          NOP
00734   740200          SZA                             /AC=0?
00735   600741          JMP     P0E0                    /NO, ERROR.
00736   440051          ISZ     CTRA                    /DONE?
00737   600730          JMP     P0TS0A                  /NO, REPEAT
00740   600260          JMP     CHAIN                   /YES, CHAIN
00741   740040  P0E0    HLT                             /ERROR HALT. DID NOT
                                                        /CLEAR AC.
00742   750001          CLA!CMA                         /-1 TO AC
00743   700311          700311                          /CLEAR AC
00744   600742          JMP .-2
00745   600742          JMP .-3
                        /ISSUE KRA, WAIT AT LEAST 200 MSEC FOR FLAG TO SET, SKIP ON FLAG.
                        /FAILURE TO SKIP INDICATES THAT FLAG IS NOT SET, OR KSF FAILURE
00746   000001  P0TS1   1
00747   000767          P0TS2
00750   100322          SETLOC                          /-200 TO DELAYM
00751   000023          DELAYM
00752   777470          -310
00753   100322  P0TS1A  SETLOC                          /-100 TO CTRA
00754   000051          CTRA
00755   777634          -144
00756   700322  P0TS1B  KRA                             /CLEAR AC AND FLAG
00757   100333          JMS     DLYMS                   /DELAY 200 MS.
00760   700301          KSF                             /SKIP IF FLAG=1
00761   600765          JMP     P0E1                    /ERROR
00762   440051          ISZ     CTRA                    /DONE?
00763   600756          JMP     P0TS1B                  /NO, REPEAT.
00764   600260          JMP     CHAIN                   /YES, CHAIN
00765   740040  P0E1    HLT                             /ERROR HALT. FLAG IS NOT 1,
                                                        /OR RSF FAILED.
00766   600753          JMP     P0TS1A                  /RESTART TEST.
                        /
                        .EJECT
```

```
                        /ISSUE KRA, WAIT AT LEAST 200 MSECS FOR FLAG TO SET, SKIP ON FLAG
                        /1000 TIMES TO VERIFY CONSISTENT SKIPPING
00767   000002  P0TS2   2
00770   001016          P0TS3
00771   100322          SETLOC          /-200 TO DELAYM
00772   000023          DELAYM
00773   777470          -310
00774   100322          SETLOC          /-1000 TO CTRA
00775   000051          CTRA
00776   776030          -1750
00777   700322  P0TS2A  KRA                             /CLEAR AC, FLAG
01000   100333          JMS     DLYMS                   /DELAY 200 MS.
01001   700301          KSF                             /READY?
01002   601010          JMP     P0E2A                   /NO, ERROR
01003   700301          KSF                             /SKIP ON FLAG=1
01004   601012          JMP     P0E2B                   /NO SKIP ERROR
01005   440051          ISZ     CTRA                    /DONE?
01006   601003          JMP     .-3                     /NO, REPEAT
01007   600260          JMP     CHAIN                   /YES, CHAIN
01010   740040  P0E2A   HLT                             /ERR A, FLAG NOT SET
                                                        /OR KSF FAILURE
01011   600777          JMP     P0TS2A
01012   740040  P0E2B   HLT                             /ERR B,KSF FAILURE
01013   700301          KSF                             /SKIP ON FLAG=1
01014   601013          JMP     .-1                     /REPEAT
01015   601013          JMP     .-2                     /REPEAT
                        /
                        .EJECT
```

```
                        /ISSUE KRA, WAIT AT LEAST 200 MSECS FOR FLAG TO SET, VERIFY THAT
                        /FLAG IS SET, RESET FLAG (KRB) AND SKIP ON FLAG 500 TIMES
                        /TO VERIFY THAT NO SKIP OCCURS WITH FLAG=0.
01016   000003  P0TS3   3
01017   001050          P0TS4
01020   100322          SETLOC                  /-200 TO DELAYM
01021   000023          DELAYM
01022   777470          -310
01023   100322          SETLOC                  /-500 TO CTRA
01024   000051          CTRA
01025   777014          -764
01026   700322  P0TS3A   KRA                     /START READ
01027   100333          JMS     DLYMS           /DELAY 200 MS.
01030   700301          KSF                     /READY?
01031   601041          JMP     P0E3A           /NO, ERROR.
01032   700312          KRB                     /YES, RESET FLAG.
01033   700301          KSF                     /READY?
01034   601036          JMP     .+2             /NO, OK
01035   601043          JMP     P0E3B           /YES, ERROR
01036   440051          ISZ     CTRA            /DONE?
01037   601033          JMP     .-4             /NO, REPEAT
01040   600260          JMP     CHAIN           /YES, CHAIN
01041   740040  P0E3A   HLT                     /ERRA, FLAG NOT SET, OR
                                                /RSF FAILED
01042   601026          JMP     P0TS3A          /TRY AGAIN
01043   740040  P0E3B   HLT                     /ERR B. FLAG FAILED TO RESET,
                                                /OR RSF SKIPPED ERRONROUSLY.
01044   700312          KRB                     /CLEAR FLAG
01045   700301          KSF                     /SKIP ON FLAG 1
01046   601044          JMP     .-2             /REPEAT
01047   601044          JMP     .-3             /REPEAT
                        /
                        .EJECT
```

```
                        /THIS ROUTINE CHECKS THAT NO OTHER DEVICE CAN CAUSE AN INTERRUPT,
                        /AND THEN CHECKS THAT THE READER IS CAPABLE OF INTERRUPTING.
01050   000004 P0TS4    4
01051   001112          P0TS5
01052   100322          SETLOC                  /SET INTERRUPT RETURN
01053   000002          2                       /TO P0E4A.
01054   001062          P0E4A
01055   703302 P0TS4A   CAF                     /CLEAR ALL FLAGS
01056   700042          ION                     /ENABLE INTERRUPT
01057   740000          NOP
01060   700002          IOF                     /DISABLE INTERRUPT
01061   601065          JMP     .+4
01062   700314 P0E4A    IORS
01063   740040          HLT                     /UNEXPECTED INTERRUPT
01064   601055          JMP     P0TS4A          /TRY AGAIN
01065   100322          SETLOC                  /-1000 TO CTRA
01066   000051          CTRA
01067   776030          -1750
01070   100322          SETLOC                  /SET INTERRUPT RETURN
01071   000002          2                       /TO P0TS4C
01072   001107          P0TS4C
01073   700322          KRA                     /START READER
01074   700301          KSF                     /WAIT FOR FLAG1
01075   601074          JMP     .-1
01076   700042 P0TS4B   ION                     /ENABLE INT.
01077   740000          NOP
01100   740040 P0E4B    HLT                     /READER FAILED TO INTERRUPT
                                                /OR INTERRUPT SYSTEM MALFUNCTION.
01101   100322          SETLOC                  /SET INTERRUPT RETURN
01102   000002          2                       /TO P0TS4C-1
01103   001106          P0TS4C-1
                        /SCOPE LOOP
01104   700042          ION
01105   740000          NOP
01106   601104          JMP     .-2
                        /
01107   440051 P0TS4C   ISZ     CTRA            /DONE?
01110   601076          JMP     P0TS4B          /NO, REPEAT
01111   600260          JMP     CHAIN           /YES, CHAIN
                        /
                        .EJECT
```

```
                        /READER TIMING TEST. CHECKS THAT READER FLAG IS=1 NO LATER THAN
                        /110 MILLISECONDS AFTER RSA INSTRUCTION IS ISSUED.
01112   000005  P0TS5    5
01113   001136          P0TS6
01114   100322          SETLOC                          /-110 TO DELAYM
01115   000023          DELAYM
01116   777622          -156
01117   100322          SETLOC              /-100 TO CTRA
01120   000051          CTRA
01121   777634          -144
01122   700322  P0TS5A   KRA                             /START READER CLEAR AC AND FLAG
01123   100333          JMS         DLYMS               /DELAY 110 MSECS
01124   700301          KSF                             /READY?
01125   601131          JMP         P0E5                /NO, ERROR
01126   440051          ISZ         CTRA                /DONE?
01127   601122          JMP         P0TS5A              /NO, REPEAT
01130   600260          JMP         CHAIN               /YES, CHAIN
01131   740040  P0E5     HLT                            /ERROR, FLAG NOT 1 AFTER
                                                        /110 MSECS.
01132   700322           KRA                            /START READER CLEAR
01133   700301           KSF                            /READY?
01134   601133           JMP         .-1                /NO.
01135   601132           JMP         .-3                /YES, START READER AGAIN
                         .EJECT
```

```
                        /READS 256 DIFFERENT CHARACTERS, EACH CHARACTER IS REREAD 1000 TIMES
                        /TO VERIFY CONSISTENCY IN READING FROM TTI.
01136   000006  P0TS6   6
01137   777777          777777
01140   100322          SETLOC          /-256 TO CTRA
01141   000051          CTRA
01142   777400          -400
01143   703302  P0TS6A  CAF             /CLEAR FLAGS
01144   700322          KRA                     /START READER CLEAR
01145   700301          KSF                     /READY?
01146   601145          JMP     .-1             /WAIT
01147   700312          KRB                     /READ CHAR
01150   041177          DAC     WTS6A                   /SAVE IT
01151   100322          SETLOC                  /-1000 TO CTRB
01152   000052          CTRB
01153   776030          -1750
01154   700312  P0TS6B  KRB
01155   041200          DAC     WTS6B           /SAVE
01156   740001          CMA
01157   340061          TAD     K1
01160   341177          TAD     WTS6A
01161   750200          SZA!CLA                 /RESULT 0?
01162   601170          JMP     P0E6A           /NO, ERROR.
01163   440052          ISZ     CTRB            /REREAD 1000 TIMES?
01164   601154          JMP     P0TS6B          /NO GO READ AGAIN
01165   440051  P0TS6C  ISZ     CTRA            /READ 256 CHARACTERS?
01166   601143          JMP     P0TS6A          /NO
01167   600260          JMP     CHAIN           /YES, CHAIN
01170   201200  P0E6A   LAC     WTS6B           /LOAD BAD CHAR
01171   740040          HLT                     /ERR A. BAD CHAR IN AC
                                                /PRESS CONTINUE.
01172   201177          LAC     WTS6A           /EXPECTED CHAR IN AC
01173   740040          HLT
01174   601165          JMP P0TS6C
                        /SCOPE LOOP
01175   700312          KRB                     /READ CHARACTER
01176   601175          JMP     .-1             /REPEAT
01177   000000  WTS6A   0
01200   000000  WTS6B   0
                        /
                        .EJECT
```

```
                        /PROGRAM 1. ASR33/35 TELETYPE BASIC OUTPUT TESTS. PROGRAM CHECKS
                        /OUTPUT IOT'S, INTERRUPT, TIMING.
01201   100322  PRG1    SETLOC                  /SET KSTART TO INITIAL
01202   000020            KSTART                /ROUTINE ADDRESS
01203   001205            P1TS0
01204   600233            JMP     SRSET         /START PROGRAM
                        /
                        /ROUTINE 0
                        /1. TLS AND WAIT AT LEAST 200 MSECS FOR FLAG TO SET. SKIP ON FLAG=1 (TSF). TSF SHOU
                        /SKIP OR ERROR HALT P1E0A OCCURS, INDICATING FLAG NOT SET, OR TSF FAILURE.
                        /2. WITH FLAG=1, SKIPS ON FLAG 1000 TIMES TO TEST FOR CONSISTENT SKIPPING.
                        /FAILURE TO SKIP CAUSES ERROR HALT P1E0B.
01205   000000  P1TS0     0
01206   001235            P1TS1
01207   100322            SETLOC                /-200 TO DELAYM
01210   000023            DELAYM
01211   777470            -310
01212   100322            SETLOC                /-1000 TO CTRA
01213   000051            CTRA
01214   776030            -1750
01215   750000  P1TS0A    CLA                   /CLEAR AC
01216   700406            TLS                   /START PRINTER/PUNCH
01217   100333            JMS     DLYMS         /DELAY 200 MS
01220   700401            TSF                   /READY?
01221   601227            JMP     P1E0A         /NO, ERROR.
01222   700401  P1TS0B    TSF                   /SKIP ON FLAG 1
01223   601231            JMP     P1E0B         /NO SKIP. ERROR
01224   440051            ISZ     CTRA          /SKIP DONE?
01225   601222            JMP     P1TS0B        /NO, REPEAT
01226   600260            JMP     CHAIN         /YES, CHAIN
01227   750040  P1E0A     HLT!CLA               /ERR A. NOT READY AFTER 200 MS.
01230   601215            JMP     P1TS0A        /OR TSF FAILURE
01231   750040  P1E0B     HLT!CLA               /ERR B. TSF FAILED TO SKIP
                        /SCOPE LOOP
01232   700401            TSF
01233   601232            JMP     .-1
01234   601232            JMP     .-2
                        /
                          .EJECT
```

```
                        /ROUTINE 1.
                        /ISSUE TCF TO CLEAR FLAG, SKIP ON FLAG 1000 TIMES TO VERIFY THAT NO
                        /SKIP OCCURS WITH FLAG=0
01235   000001  P1TS1   1
01236   001256          P1TS2
01237   100322          SETLOC                          /-1000 TO CTRA
01240   000051          CTRA
01241   776030          -1750
01242   700402          TCF                             /CLEAR FLAG
01243   700401  P1TS1A  TSF
01244   741000          SKP
01245   601251          JMP     P1E1
01246   440051          ISZ     CTRA
01247   601243          JMP     P1TS1A
01250   600260          JMP     CHAIN
01251   750040  P1E1    HLT!CLA                         /ERR. AFTER TCF, TSF INSTRUCTION
                                                /SKIPPED.
                        /SCOPE LOOP
01252   700402          TCF                             /CLEAR FLAG
01253   700401          TSF                             /SKIP ON FLAG=1
01254   601253          JMP     .-1
01255   601253          JMP     .-2
                        /
                        .EJECT
```

```
                        /
                        /ROUTINE 2
                        /ISSUE TLS, WAIT FOR FLAG1, CLEAR FLAG (TCF), SKIP ON FLAG1, NO SKIP
                        /SHOULD OCCUR, IF SKIP, TCF INSTRUCTION (CLEAR FLAG) FAILED.
01256   000002  P1TS2   2
01257   001301          P1TS3
01260   100322          SETLOC                  /-100 TO CTRA
01261   000051          CTRA
01262   777634          -144
01263   750000  P1TS2A   CLA
01264   700406          TLS                     /PUNCH/PRINT
01265   700401          TSF                     /WAIT FOR FLAG1
01266   601265          JMP     .-1
01267   700402          TCF                     /CLEAR FLAG
01270   700401          TSF                     /SKIP IF FLAG 1
01271   741000          SKP
01272   601276          JMP     P1E2            /SKIPPED, ERROR
01273   440051          ISZ     CTRA            /DONE?
01274   601263          JMP     P1TS2A          /NO, REPEAT
01275   600260          JMP     CHAIN           /YES, CHAIN
01276   750040  P1E2    HLT!CLA                 /ERR, TCF FAILED TO RESET FLAG
                        /SCOPE LOOP
01277   700402          TCF                     /CLEAR FLAG
01300   601277          JMP     .-1             /REPEAT
                        /
                        .EJECT
```

```
                        /ROUTINE 3.
                        /THIS ROUTINE CHECKS THAT NO OTHER DEVICE CAN CAUSE AN INTERRUPT,
                        /AND THEN CHECKS THAT THE PUNCH/PRINTER CAN CAUSE AN INTERRUPT.
01301   000003  P1TS3   3
01302   001347          P1TS4
01303   100322          SETLOC              /SET INTERRUPT RETURN
01304   000002          2                   /TO P1E2A
01305   001317          P1E3A
01306   703302  P1TS3A  CAF                 /CLEAR ALL FLAGS.
01307   700406          TLS
01310   700401          TSF
01311   601310          JMP         .-1
01312   700402          TCF                 /CLEAR PUNCH/PRINTER FLAG
01313   700042          ION                 /ENABLE INT.
01314   740000          NOP
01315   700002          IOF                 /DISABLE INT.
01316   601321          JMP         .+3
01317   740040  P1E3A   HLT                 /UNEXPECTED INTERRUPT
01320   601306          JMP         P1TS3A  /TRY AGAIN
01321   100322          SETLOC              /-1000 TO CTRA
01322   000051          CTRA
01323   776030          -1750
01324   100322          SETLOC              /SET INTERRUPT RETURN
01325   000002          2                   /TO P1TS3C
01326   001344          P1TS3C
01327   750000          CLA
01330   700406          TLS                 /START PUNCH/PRINTER
01331   700401          TSF                 /FLAG 1?
01332   601331          JMP         .-1     /WAIT.
01333   700042  P1TS3B  ION
01334   740000          NOP
01335   740040  P1E3B   HLT                 /PUNCH/PRINTER FAILED TO
                                            /INTERRUPT OR INTERRUPT
                        /MALFUNCTION.
01336   100322          SETLOC
01337   000002          2
01340   001343          P1TS3C-1
01341   700042          ION                 /SCOPE LOOP
01342   740000          NOP
01343   601341          JMP         .-2
01344   440051  P1TS3C  ISZ         CTRA    /DONE?
01345   601333          JMP         P1TS3B  /NO. REPEAT
01346   600260          JMP         CHAIN   /YES. CHAIN
                        /
                        .EJECT
```

```
                        /ROUTINE 4
                        /TIMING TEST. CHECKS THAT FLAG IS 1 NO LATER THAN 110 MSECS.
                        /AFTER TLS INSTRUCTION.
01347   000004  P1TS4    4
01350   777777           777777
01351   100322           SETLOC              /-110 TO DELAYM
01352   000023           DELAYM
01353   777622           -156
01354   100322           SETLOC              /-100 TO CTRA
01355   000051           CTRA
01356   777634           -144
01357   700406  P1TS4A    TLS                 /START PUNCH/READER
01360   100333           JMS       DLYMS     /DELAY 110 MSECS.
01361   700401           TSF                 /FLAG 1?
01362   601366           JMP       P1E4      /NO. ERROR
01363   440051           ISZ       CTRA                /YES. DONE?
01364   601357           JMP       P1TS4A    /NO. REPEAT
01365   600260           JMP       CHAIN     /YES. CHAIN.
01366   750040  P1E4      HLT!CLA             /ERR. FLAG NOT 1 110 MSECS
                                             /AFTER TLS INSTRUCTION
01367   700406           TLS                 /SCOPE LOOP, START PRINTER
01370   700401           TSF                 /WAIT FOR FLAG 1
01371   601370           JMP       .-1
01372   601367           JMP       .-3                 /REPEAT.
                        /
                         .EJECT
```

```
                        /PROGRAM 2. ASR33/35 TELETYPE READER TEST. CHECKS ABILITY OF READER
                        /TO CORRECTLY READ AT FULL SPEED AND WITH RANDOM STALLS.
01373   100322  PRG2    SETLOC                  /SET KSTART TO INITIAL
01374   000020          KSTART                  /ROUTINE ADDRESS
01375   001377          P2TS0
01376   600233          JMP SRSET               /START
                        /ROUTINE 0
                        /READ 4095 CHARACTERS, AT FULL SPEED, MATCHING EACH CHARACTER READ
                        /AGAINST BINARY COUNT PATTERN
01377   000000  P2TS0    0
01400   001426          P2TS1
01401   100524          JMS     SYNK            /SYNC READER
01402   100322          SETLOC                  /-4095 TO CTRA
01403   000051          CTRA
01404   770001          -7777
01405   100406          JMS     INITPT          /INITIALIZE PATTERN
01406   100415  P2TS0A   JMS     GETPTT          /GET PATTERN CHARACTER
01407   041415          DAC     SB0             /STORE AT SB0
01410   700322          KRA                     /READ
01411   700301          KSF                     /READY?
01412   601411          JMP     .-1             /NO, WAIT
01413   700312          KRB                     /YES, READ
01414   100506          JMS     CHCK            /GO CHECK CHARACTER
01415   000000  SB0      0
01416   601422          JMP     P2E0A           /ERROR
01417   440051  P2TS0B   ISZ     CTRA            /OK. DONE?
01420   601406          JMP     P2TS0A          /NO. REPEAT
01421   600260          JMP     CHAIN           /YES. CHAIN
01422   740040  P2E0A    HLT                     /ERR HALT A. BAD CHAR IN AC.
                                                /PRESS CONTINUE.
01423   201415          LAC     SB0
01424   740040          HLT                     /CORRECT CHAR IN AC
01425   601417          JMP     P2TS0B
                        /
                        .EJECT
```

```
                        /ROUTINE 1
                        /READS 2000 CHARACTERS. RANDEM DELAY BETWEEN CHARACTERS.
                        /EACH CHARACTER READ IS MATCHED AGAINST COUNT PATTERN.
01426   000001  P2TS1   1
01427   001457          P2TS2
01430   100524          JMS     SYNK        /SYNC READER.
01431   100322          SETLOC              /-2000 TO CTRA
01432   000051          CTRA
01433   774060          -3720
01434   100406          JMS     INITPT      /INITIALIZE PATTERN
01435   100415  P2TS1A  JMS     GETPTT      /SET PATTERN CHARACTER
01436   041446          DAC     SB1
01437   700322          KRA                 /READ
01440   100474          JMS     DLCNT       /GENERATE DELAY COUNT
01441   100333          JMS     DLYMS       /DELAY
01442   700301          KSF                 /WAIT FOR
01443   601442          JMP     .-1         /READY.
01444   700312          KRB                 /READ CHARACTER
01445   100506          JMS     CHCK        /GO CHECK CHARACTER
01446   000000  SB1     0
01447   601453          JMP     P2E1A       /ERROR
01450   440051  P2TS1B  ISZ     CTRA        /DONE?
01451   601435          JMP     P2TS1A      /NO, REPEAT
01452   600260          JMP     CHAIN       /YES, CHAIN
01453   740040  P2E1A   HLT                 /ERROR HALT A. BAD CHAR IN AC
                                            /PRESS CONTINUE
01454   201446          LAC     SB1
01455   740040          HLT                 /CORRECT CHAR IN AC
01456   601450          JMP     P2TS1B
                        /
                        .EJECT
```

```
                        /ROUTINE 2
                        /READ 100 GROUPS OF CHARACTERS. EACH GROUP IS OF RANDOM LENGTH.
                        /EACH GROUP USES A FIXED DELAY DETERMINED AT RANDOM. DELAY CHANGES BETWEEN GROUPS.
01457   000002  P2TS2   2
01460   777777          777777
01461   100524          JMS     SYNK        /SYNC READER
01462   100322          SETLOC              /-100 TO CTRA
01463   000051          CTRA
01464   777634          -144
01465   100406          JMS     INITPT      /INITIALIZE PATTERN
01466   100474  P2TS2A  JMS     DLCNT       /GENERATE DELAY COUNT
01467   100455          JMS     CHRCNT      /GENERATE AND STORE
01470   000052          CTRB                /BLOCK LENGTH
01471   100415  P2TS2B  JMS     GETPTT      /GET PATTERN CHARACTER
01472   041501          DAC     SB2
01473   700322          KRA                 /READ
01474   100333          JMS     DLYMS       /DELAY
01475   700301          KSF                 /READY?
01476   601475          JMP     .-1         /NO, WAIT
01477   700312          KRB                 /READ CHARACTER
01500   100506          JMS     CHCK        /GO CHECK CHARACTER
01501   000000  SB2     0
01502   601510          JMP     P2E2A       /ERROR.
01503   440052          ISZ     CTRB        /GROUP DONE?
01504   601471          JMP     P2TS2B      /NO.
01505   440051          ISZ     CTRA        /YES. ALL GROUPS DONE?
01506   601466          JMP     P2TS2A      /NO.
01507   600260          JMP     CHAIN
01510   740040  P2E2A   HLT                 /ERR HALT A. BAD CHAR IN AC
                                            /PRESS CONTINE.
01511   201501          LAC     SB2
01512   740040          HLT                 /CORRECT CHAR IN AC
01513   601503          JMP     SB2+2
                        /
                        .EJECT
```

```
          000000                .END
01514     000405    *L
01515     000375    *L

                                NO ERROR LINES
```